
Récupération d'une imprimante Laser si bloquée en pending 
 

 

CMD : RCLPRT               

CMD PROMPT('Récupérer une Impr. Laser PND' 

 

PARM KWD(PRT) TYPE(*NAME) LEN(10) RSTD(*NO) +  

     MIN(1) PROMPT('Imprimante Laser (*IPDS)') 

 

 

CL : RCLPRT               

/* Compilation: avec USRPRF(*OWNER) pour Droits Adoptés */ 13/09/02 

 

PGM PARM(&PRT)  

 

DCL &PRT *CHAR (10) /* Unité */  

DCL &TYPE *CHAR 10  

DCL &TEXT *CHAR 50  

DCL &CTL *CHAR 10  

DCL &CAT *CHAR 10  

DCL &CLASSE *CHAR 10  

 

DCLMSG: DCL &MSG *CHAR 80 /* Standard Gestion des erreurs */  

DCL &MSGF *CHAR 10  

DCL &MSGFLIB *CHAR 10  

DCL &MSGID *CHAR 7  

DCL &MSGDTA *CHAR 80  

 

DCL &RTNTYPE *CHAR 2  

DCL &DIAG *CHAR 2 '02'  

DCL &EXCP *CHAR 2 '15'  

DCL &CANCELD *CHAR 30 'Commande RCLPRT annulée'  

 

/* ----------------------------------------------------------------- */  

/* Moniteur général d'erreurs et routine de gestion des erreurs */  

MONMSG MSGID(CPF0000 MCH0000) EXEC(GOTO ERREUR)  

/* ----------------------------------------------------------------- */  

 

RTVDEVD DEVD(&PRT) CAT(&CAT) TYPE(&TYPE) +  

  DEVCLS(&CLASSE) TEXT(&TEXT) CTL(&CTL)  

 

IF (&TYPE ˆ= '*IPDS') DO  

CHGVAR VAR(&MSG) VALUE('L''imprimante ' *CAT &PRT +  

*BCAT 'n''est pas de type *IPDS..')  

SNDPGMMSG MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA(&MSG) +  

    TOPGMQ(*PRV) MSGTYPE(*COMP)  

GOTO FINPGM  

ENDDO  

IF (&CTL = ' ') CHGVAR &CTL &CLASSE  

CHGVAR VAR(&MSG) VALUE('Arrêt de ' *CAT &PRT *BCAT +  

'sur ' *CAT &CTL *BCAT 'en cours..')  

SNDPGMMSG MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA(&MSG) +  

    TOPGMQ(*EXT) MSGTYPE(*STATUS)  

 

ENDWTR WTR(&PRT) OPTION(*IMMED)  

DLYJOB DLY(5)  

 

CHGVAR VAR(&MSG) VALUE('Récupération de ' *CAT &PRT +  

 *BCAT '/' *BCAT &TEXT *BCAT 'en cours..')  

SNDPGMMSG MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA(&MSG) +  

    TOPGMQ(*EXT) MSGTYPE(*STATUS)  

 

CALL QSPFIXUP  

 

CHGVAR VAR(&MSG) VALUE('Redémarrage de ' *CAT &PRT +  

 *BCAT 'sur ' *CAT &CTL *BCAT 'en cours..')  

SNDPGMMSG MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA(&MSG) +  

    TOPGMQ(*EXT) MSGTYPE(*STATUS)  

 

STRPRTWTR DEV(&PRT)  

DLYJOB DLY(5)  

 

CHGVAR VAR(&MSG) VALUE('L''imprimante ' *CAT &PRT +  

 *BCAT 'doit être à nouveau opérationnelle..')  

SNDPGMMSG MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA(&MSG) +  

    TOPGMQ(*PRV) MSGTYPE(*COMP)  

 

RETURN  

ERREUR:  

/* Réception des Messages */  

RCVMSG: RCVMSG MSGDTA(&MSGDTA) MSGID(&MSGID) +  

   RTNTYPE(&RTNTYPE) MSGF(&MSGF) MSGFLIB(&MSGFLIB)  

 

/* Renvoi Messages */  

IF COND(&MSGID *NE ' ') THEN(DO)  

 

IF ((&RTNTYPE = &DIAG) *OR (&RTNTYPE = &EXCP)) +  

DO  

SNDPGMMSG MSGID(&MSGID) MSGF(&MSGFLIB/&MSGF) +  

    MSGDTA(&MSGDTA) TOPGMQ(*PRV) MSGTYPE(*ESCAPE)  

MONMSG CPF0000  

ENDDO  

 

GOTO RCVMSG  

ENDDO  

 

ELSE DO  

/* Message de Fin anormale */  

SNDPGMMSG MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA(&CANCELD) +  

    TOPGMQ(*PRV) MSGTYPE(*ESCAPE)  

MONMSG CPF0000  

ENDDO  

FINPGM: 

ENDPGM 

 


